Technical paper 5

Custom data charts

Last amended: 25th February 2001

Map Maker can be configured to draw customised data charts (see Part 2 – chapter 5). To use this facility you must create a Dynamic Link Library (DLL) to process the data sent by Map Maker for each object.

The data for each object

As the map is drawn by Map Maker, each time a custom chart is to be drawn for an object the program generates a short text file containing the data that needs to be displayed on the chart for that particular object. This text file is sent to the custom DLL. The text file is in the form of a configuration file (*.ini) with one section called "chart". For instance:

```
[chart]
type=3
caption=Zone 102
values=5
value 1=CHAFFINCH,4
value 2=BULLFINCH,0
value 3=GREENFINCH,4
value 4=GOLDFINCH,1
value 5=SISKIN,0
data colour 1=255
data colour 2=65280
data colour 3=65535
data colour 4=16711680
data colour 5=14079702
reference size=0.000000
pixels per mm=31.500000
label background=0
label bold=no
label colour=0
label font=Arial
label italic=no
label size=2.500000
```

The "type" value provides an optional value that allows the DLL to choose between drawing one of several custom charts. The "type" may be numeric, as in this case, or a string value.

The "caption" is the display label of the object in question.

"Values" states how many data values are being sent. "Value 1" to "Value n" give the name of the data field, followed by the value for that field for that particular object.

The "reference size" is an optional value which is entered by the user to determine what data value corresponds to 100%. This can be used to scale the data chart to an appropriate size.

Note that as with all Windows INI files you cannot assume that the values will appear in any particular order.

The DLL function

The DLL requires just one function. When written in Delphi it looks like:

```
function DrawChart(dc:hdc;script:shortstring):boolean;
                stdcall; export;
```

The "DC" parameter is a Windows "device context" which is the surface on which the chart is drawn. The device context, in this case, is 1000 pixels wide by 1000 pixels high. The location point of the chart is at X=500, Y=500. The top left corner is zero, zero.

Sample code

The DLL can be written in any Windows programming language. The following is sample code for Delphi 5 for a DLL to receive the data described above.

```
library CustomChart;
//generic Delphi DLL for creating a custom Data Chart
uses
  windows,SysUtils,graphics,inifiles; //Standard Delphi units


function DrawChart(dc:hdc;script:shortstring):boolean;
                stdcall; export;
 var
  canvas:Tcanvas;
  ini:Tinifile;
  NumberOfValues:integer;
  chartType:string; //developer defined chart type identifier
  caption:string;
  name:array of string; //contains the names of the fields
  value:array of string; //contains the values of the data
  n:integer;
  m:integer;
  s:string;
 begin
  result:=true;
  ini:=Tinifile.create(script);
  canvas:=Tcanvas.create;
  canvas.handle:=dc;
  with canvas do
   begin
    NumberOfValues:=ini.readInteger('chart','values',0);
    if NumberOfValues>0 then
     begin
```

```
      //get values
      setlength(name,NumberOfValues);
      setlength(value,NumberOfValues);
      for n:=1 to NumberOfValues do
       begin
        s:=ini.ReadString('chart','value '+inttostr(n),'');
        m:=pos(',',s);
        name[n-1]:=copy(s,1,m-1);
        value[n-1]:=copy(s,m+1,length(s)-m);
       end;
      caption:=ini.ReadString('chart','caption','');
      chartType:=ini.ReadString('chart','type','');
      //use chartType to determine which kind of chart to draw
      //now draw something on a 1000 x 1000 pixel canvas.
      //The insertion point for the chart is at x=500, y=500

      //e.g. draw a simple circle
      ellipse(0,0,1000,1000);

      //free dynamic data
      setlength(name,0);
      setlength(value,0);
     end;
   end;
  canvas.handle:=0; //ensure to release the device context
  canvas.free;
  ini.free;
 end;

exports

DrawChart index 1;

begin
end.
```